Computer Project 7 Help File

Part 0: Setup

Suppose that we want to examine the region that is outside the sphere $x^2 + y^2 + z^2 = 16$ and inside the paraboloid $z = 10 - (x^2 + y^2)$. We can view this region using the RegionPlot3d[...] command.

In[1]:=
```
RegionPlot3D[x^2 + y^2 + z^2 ≥ 16 && z ≤ 10 - (x^2 + y^2), {x, -5, 5},
 {y, -5, 5}, {z, 0, 10}, AxesLabel → {x, y, z}, PlotPoints → 150]
```

Out[1]=



I've rotated the image so that you can see we have a parabolic region cut off by part of a sphere. The relatively large number of plot points is to ensure that the edges aren't too pixelated. If that large of a number slows down your machine, you can probably get away with a smaller number.

If we want to do integrals over this region, we would likely want to use cylindrical coordinates. That means the sphere is given by $z = \sqrt{16 - r^2}$ while the paraboloid has the equation $z = 10 - r^2$. In order to set up an integral in cylindrical, we need to know the circle of intersection between the paraboloid and the sphere. We can easily find this by setting the two equations above equal.

In[2]:=
```
Solve[Sqrt[16 - r^2] == 10 - r^2, r]
```

Out[2]= $\{\{r \to -\sqrt{7}\}, \{r \to \sqrt{7}\}\}$

Clearly, the circle of intersection has radius $\sqrt{7}$. That means the limits of integration for our region are:

- $0 \leq \theta \leq 2\pi$,
- $0 \leq r \leq \sqrt{7}$,
- $\sqrt{16 - r^2} \leq z \leq 10 - r^2$.

To find the volume of this solid, we simply integrate one over the region. Remember to include the factor of *r* from the Jacobian for cylindrical coordinates! The volume is stored in the variable **Vol** for later use.

In[3]:=
```
Vol = Integrate[r, {θ, 0, 2*Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]
```

Out[3]= $\frac{125\pi}{6}$

Part 1: Centroid

The centroid of any region is simply the average of all the coordinate functions over the region. In effect, the centroid is just the center of mass for a region assuming that the density is constant. Since the actual value of the density is irrelevant to the location of the centroid, we can take it equal to 1.

The centroid of the region *R* is the point $(\overline{x}, \overline{y}, \overline{z})$ where

- $\overline{x} = \frac{\iiint_R x\, dV}{\text{Vol}}$
- $\overline{y} = \frac{\iiint_R y\, dV}{\text{Vol}}$
- $\overline{z} = \frac{\iiint_R z\, dV}{\text{Vol}}$

We have to convert *x*, *y*, and *z* into the appropriate quantities in cylindrical before integration (and we must also include the factor of *r* from the Jacobian). By symmetry, we expect the *x*- and *y*-coordinate of the centroid to be zero by symmetry. Of course, we can compute this directly!

In[4]:=
```
Xcent =
  Integrate[r*Cos[θ]*r, {θ, 0, 2*Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]/Vol
```

Out[4]= 0

In[5]:=
```
Ycent =
  Integrate[r*Sin[θ]*r, {θ, 0, 2*Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]/Vol
```

Out[5]= 0

In[6]:=
```
Zcent = Integrate[z * r, {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r ^ 2], 10 - r ^ 2}] / Vol
```

Out[6]= $\frac{1421}{250}$

Part 2: Center of Mass for a Given Density Function

If the region has a non-constant density, we have to adjust our computations above to find the center of mass. For our purposes, we will take the density function to be

$$\delta(x, y, z) = |x| + |y| + |z|.$$

This density function is known as the *taxicab distance* from the origin. It basically measures the total distance from any given point to each of the three coordinate axes. Essentially, this density means that regions further from the origin are more dense than regions closer to the origin. To make life easy, we will program this density as a function we can use.

In[7]:=
```
density[x_, y_, z_] := Abs[x] + Abs[y] + Abs[z]
```

The total mass of the region with this density function is given by

$$M = \iiint_R \delta(x, y, z)\, dV.$$

We will store this number in the variable **Mass** so that we can use it later. Notice that we substitute the correct expressions for x and y in cylindrical coordinates as well as include the r from the Jacobian. In addition to the exact answer, the numerical value is given below.

In[8]:=
```
Mass = Integrate[density[r * Cos[θ], r * Sin[θ], z] * r,
  {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r ^ 2], 10 - r ^ 2}]
```

Out[8]= $\frac{1}{60}\left(6856\sqrt{7} + 7105\pi - 15360\,\text{ArcSin}\left[\frac{\sqrt{7}}{4}\right]\right)$

In[9]:=
```
N[Mass]
```

Out[9]= 489.318

The center of mass for the region R is the point $(\overline{x}, \overline{y}, \overline{z})$ where

- $\overline{x} = \frac{\iiint_R x\,\delta\, dV}{\text{Mass}}$
- $\overline{y} = \frac{\iiint_R y\,\delta\, dV}{\text{Mass}}$
- $\overline{z} = \frac{\iiint_R z\,\delta\, dV}{\text{Mass}}$

In[10]:=
```
Xcom = Integrate[r * Cos[θ] * density[r * Cos[θ], r * Sin[θ], z] * r,
    {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}] / Mass
```
Out[10]= 0

In[11]:=
```
Ycom = Integrate[r * Sin[θ] * density[r * Cos[θ], r * Sin[θ], z] * r,
    {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}] / Mass
```
Out[11]= 0

In[12]:=
```
Zcom = Integrate[z * density[r * Cos[θ], r * Sin[θ], z] * r,
    {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}] / Mass
```
Out[12]=
$$\frac{3\left(4704\sqrt{7}+14449\,\pi\right)}{6856\sqrt{7}+7105\,\pi-15360\,\mathrm{ArcSin}\left[\frac{\sqrt{7}}{4}\right]}$$

In[13]:=
```
N[Zcom]
```
Out[13]= 5.91011

So, the center of mass is still on the *z*-axis (because the density is still symmetric). The *z*-coordinate has shifted toward the top of the region since the regions farther from the origin are denser.

Part 3: The Inertia Tensor

Moment of Inertia is a quantity that describes how an object will behave when it is rotating around a particular axis. Since rotation is described by a point where the rotation is based and a vector giving the axis (and direction) of rotation, the moment of inertia will be different depending on the chosen axis of rotation. The object that contains this information is known as the **Inertia Tensor** which for our purposes we can think of as a matrix.

Given a point (x_0, y_0, z_0) where the tensor is centered (you can think of this as the center point of the rotation), the inertia tensor is represented 3×3 symmetric matrix given by

$$\mathbf{I} = \begin{pmatrix} I_{xx} & -I_{xy} & -I_{xz} \\ -I_{xy} & I_{yy} & -I_{yz} \\ -I_{xz} & -I_{yz} & I_{zz} \end{pmatrix}$$

where

- $I_{xx}=\int\int\int_R\left[(y-y_0)^2+(z-z_0)^2\right]\delta(x,\,y,\,z)\,dV$
- $I_{yy}=\int\int\int_R\left[(x-x_0)^2+(z-z_0)^2\right]\delta(x,\,y,\,z)\,dV$
- $I_{zz}=\int\int\int_R\left[(x-x_0)^2+(y-y_0)^2\right]\delta(x,\,y,\,z)\,dV$

- $I_{xy} = \int\int\int_R (x - x_0)(y - y_0)\,\delta(x, y, z)\,dV$
- $I_{xz} = \int\int\int_R (x - x_0)(z - z_0)\,\delta(x, y, z)\,dV$
- $I_{yz} = \int\int\int_R (y - y_0)(z - z_0)\,\delta(x, y, z)\,dV$

Typically, the best choice for the center point (x_0, y_0, z_0) is the center of mass of the object since that is the point about which an object will naturally rotate. For our purposes, we will take the density to be equal to 1. This means we should use the centroid as the center of rotation. Remember that you must change all variables to the correct expressions in the coordinates you use to integrate (and include the Jacobian).

In[14]:=
```
Ixx = Integrate[((r * Sin[θ] - Ycent)^2 + (z - Zcent)^2) * r,
  {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]
```

Out[14]= $\frac{220709\,\pi}{3000}$

In[15]:=
```
Iyy = Integrate[((r * Cos[θ] - Xcent)^2 + (z - Zcent)^2) * r,
  {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]
```

Out[15]= $\frac{220709\,\pi}{3000}$

In[16]:=
```
Izz = Integrate[((r * Cos[θ] - Xcent)^2 + (r * Sin[θ] - Ycent)^2) * r,
  {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]
```

Out[16]= $\frac{242\,\pi}{5}$

In[17]:=
```
Ixy = Integrate[(r * Cos[θ] - Xcent) * (r * Sin[θ] - Ycent) * r,
  {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]
```

Out[17]= 0

In[18]:=
```
Ixz = Integrate[(r * Cos[θ] - Xcent) * (z - Zcent) * r,
  {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]
```

Out[18]= 0

In[19]:=
```
Iyz = Integrate[(r * Sin[θ] - Ycent) * (z - Zcent) * r,
  {θ, 0, 2 * Pi}, {r, 0, Sqrt[7]}, {z, Sqrt[16 - r^2], 10 - r^2}]
```

Out[19]= 0

In[20]:=
```
MatrixForm[{{Ixx, -Ixy, -Ixz}, {-Ixy, Iyy, -Iyz}, {-Ixz, -Iyz, Izz}}]
```

Out[20]//MatrixForm=

$$\begin{pmatrix} \frac{220709\,\pi}{3000} & 0 & 0 \\ 0 & \frac{220709\,\pi}{3000} & 0 \\ 0 & 0 & \frac{242\,\pi}{5} \end{pmatrix}$$

Notice that the Inertia Tensor is a diagonal matrix here. This means that the three coordinate axes are the **principal axes of rotation** for this region. For the solid we are working with, this is not too surprising given the highly symmetric nature of the region.

Recall that every real symmetric matrix has a basis of eigenvectors. The three basis eigenvectors for the tensor **I** form the principal axes of rotation for a generic solid. The corresponding eigenvalues become the principal moments of inertia (analogous to three values above). If the inertia tensor for a solid has three distinct eigenvalues, the solid is known as an **asymmetric top.** If two of the eigenvalues are the same (as in our case), then the solid is known as a **symmetric top.** If a solid happens to have three identical eigenvalues, then it is called a **spherical top** (though the solid may not be an actual sphere).

As an another example, suppose we take a uniform cube of side-length 1. If the origin is one corner and the cube is in the first octant, then the centroid of the cube is clearly the point $\left(\frac{1}{2}, \frac{1}{2}, \frac{1}{2}\right)$ directly in the center of the cube. The inertia tensor is computed below. I decided to use C for the name of these components (C for cube).

```
In[22]:= Cxx = Integrate[(y - 1 / 2)^2 + (z - 1 / 2)^2, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[22]= $\frac{1}{6}$

```
In[23]:= Cyy = Integrate[(x - 1 / 2)^2 + (z - 1 / 2)^2, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[23]= $\frac{1}{6}$

```
In[24]:= Czz = Integrate[(x - 1 / 2)^2 + (y - 1 / 2)^2, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[24]= $\frac{1}{6}$

```
In[25]:= Cxy = Integrate[(x - 1 / 2) * (y - 1 / 2), {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[25]= 0

```
In[26]:= Cxz = Integrate[(x - 1 / 2) * (z - 1 / 2), {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[26]= 0

```
In[27]:= Cyz = Integrate[(y - 1 / 2) * (z - 1 / 2), {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[27]= 0

In[28]:=
```
MatrixForm[{{Cxx, -Cxy, -Cxz}, {-Cxy, Cyy, -Cyz}, {-Cxz, -Cyz, Czz}}]
```

Out[28]//MatrixForm=

$$\begin{pmatrix} \frac{1}{6} & 0 & 0 \\ 0 & \frac{1}{6} & 0 \\ 0 & 0 & \frac{1}{6} \end{pmatrix}$$

So, the cube is a symmetric top due its high level of symmetry.

Notice, if we compute the inertia tensor about the origin, we see something a little different.

In[31]:=
```
COxx = Integrate[y^2 + z^2, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[31]= $\frac{2}{3}$

In[30]:=
```
COyy = Integrate[x^2 + z^2, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[30]= $\frac{2}{3}$

In[32]:=
```
COzz = Integrate[x^2 + y^2, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[32]= $\frac{2}{3}$

In[34]:=
```
COxy = Integrate[x * y, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[34]= $\frac{1}{4}$

In[35]:=
```
COxz = Integrate[x * z, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[35]= $\frac{1}{4}$

In[36]:=
```
COyz = Integrate[y * z, {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
```

Out[36]= $\frac{1}{4}$

In[37]:=
```
Icube = {{COxx, -COxy, -COxz}, {-COxy, COyy, -COyz}, {-COxz, -COyz, COzz}};
MatrixForm[Icube]
```

Out[38]//MatrixForm=

$$\begin{pmatrix} \frac{2}{3} & -\frac{1}{4} & -\frac{1}{4} \\ -\frac{1}{4} & \frac{2}{3} & -\frac{1}{4} \\ -\frac{1}{4} & -\frac{1}{4} & \frac{2}{3} \end{pmatrix}$$

This inertia tensor tells you about rotational properties of the cube if the origin is the center of rotation (rather than the centroid). Even though the matrix is no longer diagonal, we can still find the principal moments and axes of rotation by looking at eigenvalues and eigenvectors. Recall that the Eigensystem[M] command gives a list of eigenvalues of M first followed by a list of the corresponding eigenvectors.

In[39]:=
```
Eigensystem[Icube]
```

Out[39]=
$$\left\{\left\{\frac{11}{12}, \frac{11}{12}, \frac{1}{6}\right\}, \{\{-1, 0, 1\}, \{-1, 1, 0\}, \{1, 1, 1\}\}\right\}$$

For rotation about a corner, the cube is no longer a spherical top! It is still a symmetric top due to the high amount of symmetry. Notice that the eigenvector corresponding to 1/6 (the smallest eigenvalue) is the one through the diagonal (the vector ⟨1, 1, 1⟩ starting at the origin). This means that the easiest way to spin a cube on its corner is to make the axis of rotation the diagonal of the cube!